

SPREADSHEETO

Functions

Review in an instant

Table of contents

Introduction	1
Foundations of functions	2
- What is a function	
- How to use any function	
Conditional functions	3
- The IF function	
- Nested IF functions	
- The IFS function	
- IF, OR + AND	
- SUMIF	
- SUMIFS	
- COUNTIF	
- COUNTIFS	
- MAXIFS and MINIFS	
Functions for working with text	11
- LEFT	
- RIGHT	
- MID	
- LEN	
- SEARCH	
- Combination of text functions	
- Joining cells	
Date functions	15
- Working with dates	
- Dissecting dates	
- Find today's date with the TODAY function	
Toolbox features	16
- Typical errors when working with functions	
- IFERROR	
- Evaluating formulas	
- Named ranges	

Introduction

This cheat sheet gives you a quick summary of all the content in "Functions".

Functions are the core of Excel. They are used to do almost any calculation in this enormous program. With more than 400 functions, it's easy to get lost. Mastering a few functions is the fastest way to master Excel. "Functions" will show you which functions you need to know, and how to use them.

This cheat sheet has multiple purposes.

Use it when faced with a specific challenge. Chances are, this cheat sheet will show you the steps on how to solve your situation. For example, if you forgot what the difference between the 'IFS function' and a 'Nested IF', you can just look it up. Or do you want to extract text from a bunch of cells? We got you covered too.

Protip: Search in the PDF for the word you're looking for. Let's say you want to know, if this cheat sheet covers "SUMIF". Then search for the word "SUMIF" and BAM, there it is on page 8.

Use this cheat sheet for when you want to review the content of this course. As time goes, you will forget a few things here and there. We all do! This cheat sheet gets you back up to speed in no time.

Don't rush the lessons, take it at a nice, slow and steady pace. You'll soon surpass friends and colleagues. One day, you'll also surpass your own wildest dreams of what you thought possible.

If you at any point have any questions, feel free to reach out to me: kasper@spreadsheeto.com

Cheers to your success!



Kasper Langmann
Co-founder
Certified Microsoft Office Specialist

Foundations of functions

What is a function?


A formula is some characters you type into a cell that contains something that is not just data. Data is just a value or a string of text, but a formula often includes an action. For example, a calculation, a reference to another cell or a function.

A function is a pre-built formula that carries out a certain calculation or action. A function has a name, like 'IF' or 'CONCATENATE', that somehow describes what it does.

Before a function can give you a result it needs input – which is called arguments. An argument can be a reference to a cell or range, a value, a string of text or even a function.

All arguments are contained between an open and a close parenthesis. All these elements are called the 'Syntax' of the function.

= FunctionName(argument, argument)



Syntax

A formula can contain one or more functions, and a function can contain one or more other functions. That could look like this:

=FunctionName(FunctionName(argument,argument),argument)

Where the function inside the other function replaces one of the arguments.


How to use any function

Many functions in Excel are very similar in the way they're used. That means that when you know a few, you know a lot.

To help you build great functions, and find the function you're looking for, Excel has a tool.

Click the 'Insert function' button either while you're trying to build your function or before you start.

This will bring up a dialog box that will help you through the syntax.



Conditional functions

The IF function

Being one of the most important functions in Excel, the IF function forms the foundation of many advanced functions.

It adds logic to your calculations. That's basically like adding a little brain to your cells that can react to the other cells, and change its output depending on what it finds.

=IF(condition, something, something else)

This edition of the syntax is my words. The real syntax goes like this:

=IF(logical_test, value_if_true, value_if_false)

But I find it more understandable to think of it as events. The first event is called 'Something'. The other event is called 'Something else'. The event that is chosen by the function depends on whether the condition (logical_text) is met or is not (is true or false).

An example of the IF function is this:

=IF(I2>110, "Large", "")

In this case, if the cell value of I2 is greater than 110, the cell with the formula will display the word "Large". If the cell value of I2 is equal to or less than 110, the cell will display nothing.

Quotation marks are used if the function is to return text, like in the above example. You can also display a value, a reference or another calculation in the 'value_if_true' or 'value_if_false' arguments. If you do, don't include the quotation marks. For example:

=IF(I2>110, 110, I2)

Nested IF functions

Nested IF functions are for when you need more than one condition in your IF function.

There are two kinds of nested IF functions.

Version1

=IF(condition, something, IF(condition, something, something else))

In this version, the second IF function (the one that is nested into the first) replaces the 'value_if_false' argument of the first IF function.

This usage of this type of nested IF function is best explained in this sentence:

If criteria1 is met, then something1 happens. If criteria1 is not met, then if criteria2 is met, something2 happens. If no criteria is met, something else happens.

Example:

=IF(I2>110, "Large", IF(I2<30, "Small", "Normal"))

If the value of cell I2 is greater than 110, then display the word "Large". If the value of cell I2 is below 30, then display the word "Small". If none of these criteria is met, then display the word "Normal". That means that the cell will contain the word "Normal" if the value in I2 is between 30 and 110.

Version2

=IF(condition, IF(condition, something, something else), something else)

In this version, the second IF function replaces the 'value_if_true' argument of the first IF function.

This usage of this type of nested IF function is best explained in this sentence:

If criteria1 is met, then if criteria2 is also met, something happens. If no criteria are met, something else happens.

Example:

=IF(I2>110, IF(H2="Products", "Large", "Not a product"), "Small")

If the value of cell I2 is greater than 110, then if the text in H2 is equal to "Product", display the word "Large".

If none of the criteria are met, display the word "Small". If the first criteria is not met, but the second criteria is, display the text "Not a product".

The IFS function

The IFS function is a new function and can only be used in Excel 2016 (which is a part of an Office 365 subscription). If you don't have Excel 2016, I recommend that you [buy it here](#). This function, and all the other functions and features that have been released with Excel 2016 are easily worth the cost of the upgrade.

With the IFS function, you can use an unlimited number of criteria for your IF statement without the hassle of nesting. This is an easier approach.

It replaces **version 1** type of nested IF function.

Syntax:

=IFS(logical_test1, value_if_true1, logical_test2, value_if_true2,...)

This function contains several logical_test and several value_if_true arguments.

This usage of this type of nested IF function is best explained in this sentence:

If a logical_test1 is true, then display value_if_true1. If logical_test2 is true, display value_if_true2.

There is no 'value_if_false' argument in this function, which means that if none of the logical_test arguments are true, the function will return an error.

Prevent that by making an extra 'logical_test' that is always true and a 'value_if_true' to display if none of the other conditions are met.

Example:

=IFS(A2=1,"Approved", A2=0,"Declined",1,"Not reviewed yet")

IF the value in A2 is equal to 1, display the text "Approved". If A2 is equal to 0, display the text "Declined". If A2 is neither 1 or 0, display the text "Not reviewed yet" (*because the 'logical_test3' is 1. And 1 is always true*).

IF + AND + OR

Together with the IF function, the AND and the OR function becomes very powerful.

They are used instead of the 'logical_test' argument in the IF function to include multiple logical tests, thus acting like a nested IF or the IFS function.

AND

This function enables you to use multiple criteria in the IF function. You type a bunch of criteria into the AND function, and if all criteria are met, the function returns 'TRUE'. If just one criteria is not met, the function will return 'FALSE'.

=AND(logical1, logical2,...)

Example:

=IF(AND(I2>110,H2="Products"), "Large", "Small")

If the value in I2 is greater than 110 and the text in H2 is equal to "Products", display the word "Large". If not, display "Small".

OR

This function enables you to use multiple criteria in the IF function. You type a bunch of criteria into the OR function, and if ONE or more criteria are met, the function returns 'TRUE'. If no criteria are not met, the function will return 'FALSE'.

=OR(logical1, logical2,...)

Example:

=IF(OR(I2>110,C2>250), "Large", "Small")

If the value in I2 is greater than 110 OR the value of C2 is greater than 250, display the word "Large". If not, display "Small".

Cheat sheet for "Functions"

SUMIF

This function sums up values given a certain criteria.

Syntax:

=SUMIF(range, criteria, [sum_range])

1. **range** = this is the range for your criteria. E.g. A2:A200.
2. **criteria** = the criteria the above range must meet. E.g. ="Products"
3. **[sum_range]** = the range you want to sum. E.g. B2:B200. Leave this out if you want to sum the range from the first criteria.

Example:

=SUMIF(A2:A200,"Products",B2:B200)

This function sums the values in column B for all the rows where the text in column A is "Products".

SUMIFS

The principle of the SUMIFS function is the same as the SUMIF function, except that with the SUMIFS function you can use multiple criteria and criteria ranges.

Syntax:

=SUMIFS(sum_range, criteria_range1, criteria1, criteria_range2, criteria2,...)

1. **sum_range** = this is the range that will be added up if the below criteria are met. E.g. D2:D200.
2. **criteria_range1** = the range that holds the cells that are tested against the criteria below. E.g. A2:A200.
3. **criteria1** = the criteria the above cells must meet to be added to the sum. E.g. "Products".
4. **criteria_range2** = another range that holds the cells that are tested against the criteria below. E.g. C2:C200.
5. **criteria2** = the criteria the above cells must meet to be added to the sum. E.g. ">="&100 → which is 'greater than or equal to 100'.

Example:

=SUMIFS(D2:D200,A2:A200,"Products",C2:C200,">="&100)

All cells in column A that is equal to the text "Products", where the value in the same row in column C is also greater than 100 are added together.

COUNTIF

This is basically the same as the SUMIF function. It just counts instead of sums.

Syntax:

=COUNTIF(range, criteria)

1. **range** = the range where you want to count some of the cells. E.g. A2:A200.
2. **criteria** = the criteria the above cells must meet to be counted.

Example:

=COUNTIF(A2:A200,"Products")

This function counts all instances of cells that are equal to the text "Products".

COUNTIFS

Using multiple criteria with the COUNTIF function is viable with the COUNTIFS function.

Syntax:

=COUNTIFS(criteria_range1, criteria1, criteria_range2, criteria2)

1. **criteria_range1** = the range where you want to count some of the cells. E.g. A2:A200.
criteria1 = the criteria the above cells must meet to be counted.
2. **criteria_range2** = the other range where you want to count some of the cells. E.g. C2:C200. The total count will be the number of rows that meet all criteria.
3. **criteria2** = the criteria the above cells must meet to be counted. E.g. ">="&100 → which is 'greater than or equal to 100'.

Example:

=COUNTIFS(A2:A200,"Products",C2:C200,">="&100)

This function counts all instances of cells that are equal to the text "Products" where the value in column C in the same row is equal to or greater than 100.

MAXIFS and MINIFS

The MAX and MIN functions return the highest and lowest value in a range.

The MAXIFS and MINIFS functions allow you to add a set of conditions to these functions. The function then finds the maximum or minimum value only for the values that meet the condition(s).

Syntax:

=MAXIFS(max_range, criteria_range1, criteria1,...)

1. **max_range** = the range where you want to find the maximum value. E.g. B2:B200.
2. **criteria_range1** = the range with cells that must meet the criteria. The max_range will exclude values in the rows where the criteria_range does not meet the criteria below. E.g. A2:A200.
3. **criteria1** = the criteria for the criteria_range. E.g. ">"&500.

For more criteria, use criteria_range2 + criteria2... as with SUMIFS and COUNTIFS above.

Example:

=MAXIFS(B2:B200,A2:A200,">"&500)

Find the maximum value of all values in B2:B200, if the value in column A in the same row is greater than 500.

MINIFS

To use the MINIFS function, simply change the function name to MINIFS instead of MAXIFS.

Example:

=MINIFS(B2:B200,A2:A200,"<"&500)

Find the minimum value of all values in B2:B200, if the value in column A in the same row is less than 500.

Functions for working with text

LEFT

Is for retrieving characters from a string of text from the left side of a cell.

Syntax:

=LEFT(text, [num_chars])

1. **text** = a reference to the cell that holds the text you want to retrieve characters from. E.g. A2.
2. **[num_chars]** = the number of characters you want to retrieve. E.g. 4.

Example:

=LEFT(A2,4)

This function returns the first 4 characters of cell A2. If the cell A2 contains the text "John Davis", the function will return "John".

RIGHT

Is for retrieving characters from a string of text from the right side of a cell.

Syntax:

=RIGHT(text, [num_chars])

1. **text** = a reference to the cell that holds the text you want to retrieve characters from. E.g. A2.
2. **[num_chars]** = the number of characters you want to retrieve. E.g. 5.

Example:

=RIGHT(A2,5)

This function returns the last 5 characters of cell A2. If the cell A2 contains the text "John Davis", the function will return "Davis".

MID

Is for retrieving characters from a string of text from anywhere in a cell.

Syntax:

=MID(text, [num_chars])

1. **text** = a reference to the cell that holds the text you want to retrieve characters from. E.g. A2.
2. **start_num** = where you want to retrieve characters from. E.g. 6 (the 6th position in the cell).
3. **num_chars** = the number of characters you want to retrieve. E.g. 5.

Example:

=MID(A2,6,5)

This function returns 5 characters of cell A2 starting from (and including) position 6. If the cell A2 contains the text "John Davis", the function will return "Davis".

This function returns the first 8 characters of cell A2. If the cell A2 contains the text "John Davis", the function will return "Davis".

LEN

This function counts how many characters a cell contains.

Syntax:

=LEN(text)

1. **text** = a reference to the cell that holds the number of characters you want to count.

Example:

=LEN(A2)

This function counts the characters in cell A2. If the cell A2 contains the text "John Davis", the function will return the number 10.

SEARCH

This function returns the position of a search inquiry (a string of text or a character) in a cell.

Syntax:

=SEARCH(find_text, within_text, [start_num])

1. **find_text** = this is the character or string you're searching for. E.g. " " (blank space).
2. **Within_text** = a reference to the cell you're searching in. E.g. A2.
3. **[start_num]** = the position you want to start the search from. Only used when you know the 'within_text' contains more than one instance of the 'find_text'. This is optional, so it's often omitted.

Example:

=SEARCH(" ",A2)

This function finds the space in A2. If there are none, the function will return an error. If there is a space, the function will return its position. If A2 contains "John Davis", the function will return the number 5.

Combination of text functions

In the real world, when extracting first names and last names, you never know the position of the space. And the space separates the first name from the last name.

To separate names, you need to combine several text functions.

Separate first names

A2 contains a full name like "Catherine Gonzales"

Use this function to extract the first name, no matter how many characters it consists of.

=LEFT(A2,SEARCH(" ",A2)-1)

Separate last names

Use this function to extract the first name, no matter how many characters it consists of.

=RIGHT(A2,LEN(A2)-SEARCH(" ",A2)-1)

Joining cells

Joining the content of cells is a very handy feature of Excel. There are multiple ways of doing this. Here are the two most popular methods.

&

Put the two cells in a formula. If you want to join cell A2 and B2, use this formula:

`=A2&B2`

If you want a space between the content of the two cells, use this formula:

`=A2&" "&B2`

CONCATENATE

This function is well known for its ability to join the content of multiple cells.

Syntax:

`=CONCATENATE(text1, [text2],...)`

1. **text1** = One of the cells or strings of text you want to join.
2. **[text2]** = Any other cell or string of text you want to join.

Example:

`=CONCATENATE(A2," ",B2)`

If the cell A2 contains "Rick" and cell B2 contains "Harris", this function will return "Rick Harris".

Date functions

Working with dates

It's very important to understand that dates in Excel are just numbers formatted as dates.

That means that the number 1 formatted as a date is 1/1/1900. The number 20,000 formatted as a date is 10/3/1954.

To subtract two dates, you subtract two numbers.

Number of days between two dates:

=Latest date – earliest date

Dissecting dates

Find the day/month/year of a date with the DAY/MONTH/YEAR function.

Syntax:

=DAY(serial_number)

=MONTH(serial_number)

=YEAR(serial_number)

1. **serial_number** = this single argument is just the date you want to retrieve information from. E.g. A2 (a cell reference to the date).

Example:

=MONTH(A2)

Finds the number value of the month from the date in cell A2. The date is 19/2/2005, so the function returns the number 2.

Using today's date in Excel

Today's date can be found with the TODAY function in Excel. This function updates itself all the time, so will always show the current date.

The syntax for the TODAY function is this (the function has no arguments).

=TODAY()

Example:

=TODAY()

The function returns the serial number for 11/16/2016 because today is 11/16/2016. The serial number is then automatically formatted as a date.

Toolbox features

Typical errors

Errors are quite normal when you work with functions (and many other features) in Excel.

The most common types of errors and their causes are listed here:

Error type	Happens
#VALUE!	When types of data are not matching
#REF!	When a cell reference is deleted or moved
#DIV/0!	When trying to divide with zero
#NAME?	When mistyping references or function names

IFERROR

With the IFERROR function, you can choose what happens when a formula returns an error.

You build the IFERROR function around your original formula.

Syntax:

=IFERROR(value, value_if_error)

1. value = this is the original formula. E.g. =B100+C100/D100
2. value_if_error = this is what happens if the above formula returns an error. E.g. displays the text "Can't find".

Example:

=IFERROR(B100+C100/D100,"")

This function hides the ugly error messages if the original formula returns an error. The "" means that nothing is displayed if the 'value' argument returns an error.

Evaluating formulas

When using conditional functions, you can sometimes lose track of it when it returns an error or shows the wrong result. You need to check each argument to find the error and to understand how it works.

The F9 shortcut helps you with this.


Go into a function and use the tooltip(!) to select an entire logical test.

Hit F9 on your keyboard and the logical test evaluates into one of two results – TRUE or FALSE.

Continue to do so for all elements of the syntax that you're confused with.

This will give you a much greater understanding of the formula.


When you're done evaluating, don't hit Enter as it will apply all the changes to your formula. That will make the formula not work properly. Instead hit Esc to cancel the changes.



Named ranges

You can give new names to the ranges you use in your formulas. This way, you can refer to ranges much easier than normally.

To name a range, select the range you want to name and enter the new name into the box left of the formula bar.



Now, when creating a formula, use this name. For example:

`=INDEX(TotalPrices,MATCH(A3,ItemNo,0))`